ABSTRACT

Data from US federal health surveys frequently use complex survey structures, rendering traditional procedures not useful for analysis. The SAS survey procedures exist but have not yet become a regularly used asset in analysis. Instead, users frequently choose to use other programs or add-ons for even the most basic of analyses. This paper demonstrates why the survey procedures such as SURVEYFREQ, SURVEYLOGISTIC, and SURVEYREG should be in everyone’s toolbox when using complex survey data in research or practice.

INTRODUCTION

Categorical data analysis includes, but is not limited to, calculating basic frequencies of categorical variables, analyzing the simple relationships between two variables, and analyzing the relationships between multiple variables at the same time. No matter which of these situations you are faced with, it is extremely important to know what SAS® tools to use and how to interpret the results from those tools. An important factor to consider is whether the data was created with multi-stage probability sampling and whether it is intended to be used as such.

Multi-stage probability sampling is the combined use of various sampling methods, including cluster sampling and stratified sampling, to collect information on multiple segments of a population in order to ensure truly random but representative selection. Weights can be applied to this data to provide regional or national estimates while taking into account known distributions of age, sex, race, etc. For this paper, we will refer to this entire sampling process as complex survey sampling. Many surveys in the United States, such as the Behavioral Risk Factor Surveillance System (BRFSS) and the National Health and Nutrition Examination Survey (NHANES), utilize complex survey sampling to produce data that represents the health status of the entire United States.

Table 1 lists SAS procedures that may be very familiar to SAS users (non-survey procedures) and can often use weights. It also includes procedures (survey procedures) that can take advantage of all the components of complex survey samples including strata and clusters.

<table>
<thead>
<tr>
<th>Non-Survey Procedures</th>
<th>Survey Procedures</th>
</tr>
</thead>
<tbody>
<tr>
<td>PROC FREQ</td>
<td>PROC SURVEYFREQ</td>
</tr>
<tr>
<td>PROC REG</td>
<td>PROC SURVEYREG</td>
</tr>
<tr>
<td>PROC LOGISTIC</td>
<td>PROC SURVEYLOGISTIC</td>
</tr>
<tr>
<td>PROC MEANS</td>
<td>PROC SURVEYMEANS</td>
</tr>
<tr>
<td>PROC PHREG</td>
<td>PROC SURVEYPHREG</td>
</tr>
<tr>
<td></td>
<td>PROC SURVEYSELECT</td>
</tr>
<tr>
<td>PROC MI/PROC MIANALYZ</td>
<td>PROC SURVEYIMPUTE</td>
</tr>
</tbody>
</table>

Table 1. SAS Survey and Non-Survey Procedures

Of the procedures listed in Table 1, several are useful for categorical data analysis. This paper will specifically discuss some considerations for writing and running the SURVEYFREQ procedures, the SURVEYLOGISTIC procedure, and the SURVEYREG procedure code, as well
as important output to pay attention to as you implement these procedures in your research.

DATA AND SCENARIO

To explain and demonstrate the procedures in this paper, we will use the following data and scenario. We are conducting a study using the 2017 BRFSS data from the Centers for Disease Control and Prevention (CDC). This data is freely available and the location can be found in the References section of this paper.

To obtain the best results for data that comes from complex survey data, we start with the complete original data set. The data set should include variables representing the strata, weights, replicate weights, and/or clusters that were utilized in the sampling. Some data may not include all four of these, but if they are present they should be taken into account. It is acceptable to delete unneeded variables but all observations should be retained. Doing this ensures that any statistics will appropriately take advantage of the sampling design. An appropriate sample, if needed, can be obtained by using the SURVEYSELECT procedure.

To create the data for this paper, we used the following syntax:

```sas
data brfss2;
set brfss;
keep _age_g sex wtkg3 persdoc2 _rfbmi5 cvdcrhd4 _race_g1
    _psu _STSTR _LLCPWT;
    wtkg3 = wtkg3*0.01;
    if sex = 9 then sex = .;
    if persdoc2 in (7,9) then persdoc2 = .;
    if persdoc2 in (1,2) then persdoc2 = 1;
    if _rfbmi5 = 9 then _rfbmi5 = .;
    if cvdcrhd4 in (7,9) then cvdcrhd4 = .;
run;
```

We are interested in knowing a) whether there is a relationship between heart disease and access to care, b) whether that relationship is different for people who are normal or underweight compared to those that are overweight or obese, and c) how heart disease and access to care influence weight. To make sure we can use this data to answer our questions, we used the codebook from BRFSS 2017 to identify what variables might be useful. The variables _psu, _ststr, and _llcpwt are noted in the BRFSS 2017 documentation to be the cluster, strata, and weight variables respectively. The other variables being kept in this data step are our variables of interest – heart disease (cvdcrhd4), a marker of BMI level (_rfbmi5), a marker of weight (WTKG3), and a marker of access to care (persdoc2) – and other variables that could be potentially used in analysis (age, sex, race). The heart disease variable is a yes/no variable. The BMI variable is two levels – underweight/normal weight and overweight/obese. The access to care variable is an indicator of whether someone has a doctor or not and is also a yes/no variable. The modifications being made using the IF-THEN statements are only being used to create two level variables for the example analyses. Instructions in the codebook indicate that the variable WTKG3 has 2 implied decimals. To insert these decimals, we multiply WTKG3 by 0.01.

ENTER THE SURVEY PROCS!

Using the SAS survey procedures is not drastically different than using procedures that SAS users use for non-complex survey data. An overall piece of guidance is that for the survey procedures there can only be one WEIGHT statement. However, multiple STRATA, replicate weights (REPWEIGHTS), or CLUSTER statements can be used in the same procedure. If
using REPWEIGHTS, STRATA and CLUSTER statements will be ignored. Let’s review basics of our three procedures through five examples.

**DIFFERENCES IN THE PROC SURVEYFREQ AND PROC FREQ CODE**

The basic structure of PROC SURVEYFREQ code has some similarities to PROC FREQ, but also has several key differences:

**PROC FREQ**

- `<options>`;
- `BY variables`;
- `EXACT statistic-options < / computation-options > ;`;
- `OUTPUT <OUT=SAS-data-set > output-options ;`;
- `TABLES requests < / options > ;`;
- `TEST options`;
- `WEIGHT variable < / option > ;`;
- `RUN`;

**PROC SURVEYFREQ**

- `<options> ;`
- `BY variables;
- `CLUSTER variables;
- `REPWEIGHTS variables < / options > ;
- `STRATA variables < / option > ;`
- `TABLES requests < / options > ;`;
- `WEIGHT variable;
- `RUN`;

The differences between the FREQ procedure and PROC SURVEYFREQ are highlighted in yellow above. As we have discussed, PROC SURVEYFREQ takes into account sampling clusters and strata that PROC FREQ cannot, ensuring that standard errors are accurate. This is the primary reason for using PROC SURVEYFREQ instead of PROC FREQ. The only required statements for either procedure are the PROC statements and RUN. It is good practice to specify the data set the procedures are to use. If more than one-way tables are necessary or we only need information on specific variables, the TABLES statement is also required.

**EXAMPLE 1**

To begin our study, we are interested in finding out how many residents of the United States have ever been diagnosed with heart disease and how many are underweight/normal weight or overweight/obese. We are also interested in finding out how many people have access to care.

If we were to use PROC FREQ and the weight variable `_llcpwt` our code might look like this:

```bash
proc freq data = brfss2;
tables _rfbmi5 cvdcrhd4 persdoc2;
weight _llcpwt;
run;
```

Output 1 contains the results from PROC FREQ:
If we were to use PROC SURVEYFREQ and the weight (_llcpwt), strata (_ststr), and cluster (_psu) variables, our code might look like this:

```plaintext
proc surveyfreq data = brfss2;
  cluster _psu;
  strata _ststr;
  weight _llcpwt;
  tables _rfbmi5 cvdcrhd4 persdoc2;
run;
```

Output 2 contains the results from PROC SURVEYFREQ:
Output 2. Obesity, Heart Disease, and Access to Care Weighted Frequency Output from PROC SURVEYFREQ

While the weighted frequencies obtained from PROC FREQ (Output 1) and PROC SURVEYFREQ (Output 2) appear to be equal, we know those obtained from PROC SURVEYFREQ are more accurate for the nationally weighted estimate because they take into account the strata and clusters. The “Sum of Weights” in the first table of results from PROC
SURVEYFREQ (Output 2) gives the total weighted population size of 255,653,205 people whereas the total from PROC FREQ does not total this same amount even though it is close.

**DIFFERENCES IN THE PROC SURVEYLOGISTIC AND PROC LOGISTIC CODE**

The differences between the LOGISTIC procedure and PROC SURVEYLOGISTIC are highlighted in yellow below.

---

**PROC LOGISTIC** <options>;
BY variables;
CLASS variable <(options)> <variable <(options)> ...> </ options>;
CODE <options>;
CONTRAST 'label' effect values<, effect values, ...> </ options>;
EFFECT name=effect-type(variables </ options>);
EFFECTPLOT <plot-type <(plot-definition-options)>> </ options>;
ESTIMATE 'label' estimate-specification </ options>;
EXACT '<label>' <INTERCEPT> 'effects' </ options>;
EXACTOPTIONS options;
FREQ variable;
ID variables;
LSMEANS <model-effects> </ options>;
LSMESTIMATE model-effect lsestimate-specification </ options>;
MODEL variable <(variable_options)> = <effects> </ options>;
MODEL events/trials = <effects> </ options>;
NLOPTIONS options;
ODDSRATIO '<label>' variable </ options>;
OUTPUT <OUT=SAS-data-set> <keyword=name <keyword=name ...>> </ option>;
ROC '<label>' <specification> </ options>;
ROCCONTRAST '<label>' <contrast> </ options>;
SCORE <options>;
SLICE model-effect </ options>;
STORE <OUT=item-store-name </ LABEL='label'>;
STRATA effects </ options>;
TEST equation1 <,equation2, ...> </ option>;
UNITS <independent1=list1 <independent2=list2 ...>> </ option>;
WEIGHT variable </ option>;
RUN;

**PROC SURVEYLOGISTIC** <options>;
BY variables;
CLASS variable <(v-options)> <variable <(v-options)> ...> </ v-options>;
CLUSTER variables;
CONTRAST 'label' effect values<, effect values, ...effect values> </ options>;
DOMAIN variables <variable*variable variable*variable variable*variable ...>;
EFFECT name=effect-type (variables </ options>);
ESTIMATE '<label' estimate-specification </ options>;
FREQ variable;
LSMEANS <model-effects> </ options>;
LSMESTIMATE model-effect lsestimate-specification </ options>;
MODEL events/trials = <effects> </ options>;
MODEL variable <(v-options)> = <effects> </ options>;
OUTPUT <OUT=SAS-data-set> <options> </ option>;
REPWEIGHTS variables </ options>;
SLICE model-effect </ options>;
STORE <OUT=item-store-name </ LABEL='label'>;
STRATA variables </ option>;

---
PROC SURVEYLOGISTIC takes into account sampling clusters and strata that PROC LOGISTIC cannot. The only required statements for either procedure are the PROC statement, MODEL statement, and RUN. It is good practice to specify the data set the procedures are to use. If using CLASS or EFFECT statements they must come before the MODEL statement. DOMAIN is a statement present in several of the survey procedures but not in PROC SURVEYFREQ. We use it instead of the BY statement to do analyses on particular subgroups of the population. The DOMAIN statement allows you to get results for each level of a specific variable the same way you would with BY. If you were only interested in results for one group (for example males), you would only pay attention to those results and ignore the results for the other groups or levels present in the DOMAIN variable.

**EXAMPLE 2**

We now want to look at the relationship between heart disease and access to care. We can use PROC SURVEYFREQ or PROC SURVEYLOGISTIC for this.

Our PROC SURVEYFREQ code might look like this:

```plaintext
proc surveyfreq data = brfss2;
   cluster _psu;
   strata _ststr;
   weight _llcpwt;
   tables persdoc2*cvdcrhd4 / or;
run;
```

Just as with PROC FREQ, we can use options such as OR to obtain relative risk and odds ratio results for our analyses.

Output 3 contains the results from PROC SURVEYFREQ:
Output 3. Simple Relationship Between Heart Disease and Access to Care Output from PROC SURVEYFREQ

Unlike the default PROC FREQ output, PROC SURVEYFREQ does not provide an N-way (2x2 in this case) table with row and column percentages for the results. However, it does include the weighted frequencies and percentages for each group so one can obtain the exact same data as needed. Just as with PROC FREQ, PROC SURVEYFREQ allows the use of options to obtain statistics such as the odds ratio for looking at the relationship between two variables. In this example, we find that people in the US who have at least one doctor are 3.79 times more likely to have been diagnosed with heart disease than people who have no doctor.

Remembering that we can also assess this relationship with a simple logistic regression, our PROC SURVEYLOGISTIC code might look like this:

```plaintext
proc surveylogistic data = brfss2;
cluster _psu;
strata _ststr;
```
weight _llcpwt;
class persdoc2 (ref='No Doc');
model cvdcrhd4 (event='Yes')= persdoc2;
run;

Output 4 contains selected results from the PROC SURVEYLOGISTIC code.

Output 4. Simple Relationship Between Heart Disease and Access to Care Output from PROC SURVEYLOGISTIC

The fourth table in the PROC SURVEYLOGISTIC results is very important (the first table in Output 4). It specifically lets you know what outcome your code is written to model. In this case, we are modeling the probability of having been diagnosed with heart disease. The (event = 'Yes') syntax on the model statement was used to make sure this happened. Without it, SAS defaulted to ‘No’ as the outcome to be modeled.

Since we obtained an odds ratio from PROC SURVEYFREQ, we can compare that value to the information in the eleventh table in the PROC SURVEYLOGISTIC results (the second table in Output 4). Just as before, we find that people who have at least one doctor are 3.79 times more likely to have been diagnosed with heart disease than people who have no doctor. It is worth noting here that we ensured that our comparison would be ‘At Least One Doc’ vs ‘No Doc’ by including the CLASS statement to tell SAS that persdoc2 was a categorical variable and the (ref='No Doc') option in that same statement to tell SAS what the reference group should be.

EXAMPLE 3

The next step in this example study is to determine if the relationship found in Example 2 is different for people who are normal/underweight compared to those that are overweight/obese. As with Example 2, we can use both PROC SURVEYFREQ and PROC SURVEYLOGISTIC.

If we were using PROC FREQ instead of PROC SURVEYFREQ, we would likely use a BY statement to ask for results for our two groups of BMI from our variable _rfbmi5. Because we are using our complex survey data, we need to use a three way table instead of a two
way table because PROC SURVEYFREQ does not have a DOMAIN statement and BY won’t handle the survey structure correctly. The three way table provides us with the best estimates that take into account our sampling structure. The first variable in the three way (_rfbmi5) is the one that contains the group(s) you want to see the two way relationship (between persdoc2 and cvdcrhd4) for. As you can see in Output 5 and Output 6, you can see that we get a table for each level of _rfbmi5 and the odds ratio for each table.

```sql
proc surveyfreq data = brfss2;
cluster _psu;
strata _ststr;
weight _llcpwt;
tables _rfbmi5*persdoc2*cvdcrhd4 / or;
run;
```

Output 5. Relationship Between Heart Disease and Access to Care for Normal/Underweight BMI Output from PROC SURVEYFREQ
Output 6. Relationship Between Heart Disease and Access to Care for Overweight/Obese BMI Output from PROC SURVEYFREQ

Our results show that people who are underweight or normal weight and have at least one doctor are 4.72 times more likely to have been diagnosed with heart disease compared to people that have no doctor. This result is different than for overweight and obese people in this same data set. People who are overweight or obese who have at least one doctor are 3.48 times more likely to have been diagnosed with heart disease compared to those that have no doctor. Based on this information, we have found the answer to our research question is that yes the relationship is different for people who are normal/underweight compared to those that are overweight/obese.

Using PROC SURVEYLOGISTIC, we can find similar answers.

```markdown
proc surveylogistic data = brfss2;
domain _rfbmi5;
cluster _psu;
strata _ststr;
weight _llcpwt;
```
We added the BMI variable to a DOMAIN statement. Whereas it was not available to use it for PROC SURVEYFREQ it is available to use here. Use of this statement continues to help us make sure to have the most accurate statistics available but, quite importantly, is how we get to see results for only the participants of interest in a study. In this example, we are interested in the results for both groups (normal/underweight and overweight/obese) but in other situations you might only be interested in the results for one of them (for example, normal/underweight). In these other situations, you only need to pay attention to the results for that level of the domain. This is best indicated in Output 7, Output 8, and Output 9.

Output 7. Output from PROC SURVEYLOGISTIC DOMAIN Summary - Normal/Underweight BMI

In Output 7 we see a box from the PROC SURVEYLOGISTIC output that specifically provides information for the subjects in the brfss2 data set that have a value of NO for being overweight or obese. These subjects are the normal/underweight persons. We can see the Sum of Weights for the Domain. This is similar as the Sum of Weights we originally saw in Output 2 – it tells us the number of observations once the aspects of the complex survey sample (weights, clusters, strata, etc) are taken into account. The results in Output 8 apply to just this group of people. The results in Output 9 apply to people who are overweight or obese.

Output 8. Relationship Between Heart Disease and Access to Care for Normal/Underweight BMI Output from PROC SURVEYLOGISTIC
As we can see in our results from the PROC SURVEYLOGISTIC (Output 8 and Output 9), we have the same results as we obtained from PROC SURVEYFREQ. If normal or underweight, people who have at least one doctor are 4.72 times more likely to have been diagnosed with heart disease compared to people that have no doctor. If obese or overweight, people who have at least one doctor are 3.48 times more likely to have been diagnosed with heart disease compared to those that have no doctor. The relationship between access to care and heart disease is different based on BMI status. This leads us to look a little closer at this relationship by advancing our use of PROC SURVEYLOGISTIC.

EXAMPLE 4

Example 3 left us with a conclusion that the primary relationship of interest between access to care and heart disease was modified by BMI status. Because of this, we want to make sure we look at our regression model again and include that information in our analysis. We do this with the addition of an interaction term, persdoc2*_rfbmi5.

```sas
proc surveylogistic data = brfss2;
cluster _psu;
strata _ststr;
weight _llcpwt;
class persdoc2 (ref='No Doc') _rfbmi5 (ref='Yes');
model cvdcrhd4 (event='Yes')= persdoc2 _rfbmi5 persdoc2*_rfbmi5;
run;
```

In this syntax, we add BMI to the CLASS statement and to the MODEL statement. We also use persdoc2*_rfbmi5 to have SAS show us the effect of third variable on the primary relationship between heart disease and access to care. We expect that term to be statistically significant in our output since we saw the difference ourselves in Example 3.
To see the effect of the third variable (BMI) on the primary relationship, we look at a different table from the PROC SURVEYLOGISTIC output. In our Output 10 we can see that BMI influences the relationship between access to care (persdoc2) and heart disease. We know this because the p-value associated with the t test for the line PERSDOC2*_RFBMI5 is less than 0.05 (if 0.05 is your significance cutoff). Because BMI is in the interaction term, it has to be in the model as a standalone variable as well. Even if it did not show as significant alone, it must stay in the model as long as the interaction term is significant. In this example, _RFBMI5 is significant with a p-value of <0.0001.

**PROC SURVEYREG**

To this point, we have discussed PROC SURVEYFREQ and PROC SURVEYLOGISTIC, two tools that are excellent for categorical data analysis with complex survey data. The REG procedure is a general all around regression tool. The benefit of it when using categorical variables is that you have the ability to look at linear (continuous) outcomes but take into consideration how groups or other qualitative information influences those outcomes. When using surveys such as the BRFSS, using the SURVEYREG procedure this can be invaluable. From the syntax below we can see that there are quite a few differences in what statements and options are available in PROC SURVEYFREQ compared to PROC REG. In PROC REG, the PROC REG statement is required. If you fit a model, as we will do, a MODEL statement is required. In PROC SURVEYREG, the PROC SURVEYREG and MODEL statements are required. The CLASS statement is required for using categorical variables. As a note, as there is no PROC SURVEYGLM, you can use PROC SURVEYREG to run models as you would in GLM. Below we can see differences in PROC REG and PROC SURVEYREG code:

```
PROC REG <options>;
<label:> MODEL dependents = <regressors> </options>;
BY variables;
FREQ variable;
ID variables;
VAR variables;
WEIGHT variable;
ADD variables;
CODE <options>;
DELETE variables;
<label:> MTEST <equation, ..., equation> </options>;
```
PROC SURVEYREG <options>;
BY variables;
CLASS variables;
CLUSTER variables;
CONTRAST 'label' effect values <...effect values> </ options>;
DOMAIN variables <variable*variable variable*variable*variable >; 
EFFECT name = effect-type (variables </ options>);
ESTIMATE '<label>' estimate-specification </ options>;
LSMEANS <model-effects> </ options>;
MODEL dependent = <effects> </ options>;
OUTPUT <keyword=variable-name> ...keyword=variable-name> </ option>;
REPWEIGHTS variables </ options>;
STORE <OUT=>item-store-name </LABEL='label'>;
STRATA variables </ options>;
TEST <model-effects> </ options>;
WEIGHT variable;
RUN;

EXAMPLE 5

The final question we wanted to answer from our scenario was whether our categorical variables for access to care and heart disease were associated with the linear version of weight. Our PROC SURVEYREG syntax might look like this:

PROC SURVEYREG data = brfss2;
CLASS persdoc2 (ref='No Doc') cvdcrhd4 (ref='No');
CLUSTER _psu;
MODEL wtkg3 = persdoc2 cvdcrhd4 / solution;
STRATA _ststr;
WEIGHT _llcpwt;
RUN;

We include our categorical variables on the CLASS statement as well as in our MODEL. As before, we make sure to set the reference groups using the ref=' ' syntax. Because we have a CLASS statement, we also want to be sure to use the option of SOLUTION on the MODEL statement to have SAS output the parameter estimates. Without it, this part of the output is suppressed. As we can see in Output 11, people who have more than one doctor are heavier by 1 kilogram than people who have no doctor when taking into consideration heart disease status. People who have heart disease are heavier by 5 kilograms than people who do not have heart disease when adjusting for access to care.
Output 11. Influence of Heart Disease and Access to Care on Weight in Kilograms - Output from PROC SURVEYREG

**COMPUTING CONSIDERATIONS**

The SAS survey PROCs can command significant computing resources, specifically RAM. They work on several versions of SAS including SAS 9.4 and SAS University Edition. Depending on the size of the data, the computer may require many or most resources that are available. If repeated slow-downs become a problem, accessing a machine with more processing power and RAM or accessing a server to run analyses may be necessary. However, in our experience, more computers are becoming available at a reasonable cost to make this even more accessible to all users.

**CONCLUSION**

PROC SURVEYFREQ, PROC SURVEYLOGISTIC, and PROC SURVEYREG are tools that should be in every programmers toolbox if they utilize complex survey sampling data. They eliminate the need for accessing additional software to do survey analysis and are capable of incorporating aspects of complex survey data such as strata, clusters, and weights that more basic procedures cannot. Spending the time to practice and incorporate the usage of these tools into a regular workflow can result in long term proficiency and continued expansion of the user base.

**REFERENCES**


**CONTACT INFORMATION**

Your comments and questions are valued and encouraged. Contact the author at:

Charlotte Baker  
Virginia Polytechnic Institute and State University  
sesug.ops.2018@gmail.com